Topologii elementare pentru AO in functionare liniara

Topologie inversoare

Scopul lucrarii

Se doreste analiza circuitului din figura:



Amplificator inversor cu AO

Componente:

R1=1kO
R2=1kO

Calculul functiei de transfer

Metoda I:divizor de tensiune

Calculul functiei de transfer folosind divisor de tensiune:

$$H(s) = -\frac{R2}{R1}$$

MetodaII:ecuatii TTN

Pentru circuitul cu nodurile din figura se scrieTTN:

(1) $V_{10}(s)=E(s)$;
(2) $-G_1V_{10}(s)+(G_1+G_2)V_{20}(s)-G_2V_{30}(s)=0$;
(3) $V_{30}(s)=-AV_{20}(s)$;
ecuatia de iesire:$V_{30}(s)=Y(s)$;

unde amplificatorul operational s-a modelat ca o sursa de tensiune comandata in tensiune.

In urma rezolvarii acestor ecuatii rezulta functia de transfer:

$$H(s) = -\frac{AR2}{AR1 + R1 + R2}$$

Metoda III: calcul simbolic

```
> restart:with(Syrup):
> libname:="C:\\maple/SCSlib",libname:
```

Caracterizarea circuitului

Descrierea circuitului folosind un netlist de tip spice

```
> AmpInversor:=
"Amplificatorul Inversor cu AO
R1 1 2 R1
R2 2 3 R2
E 3 0 0 2 A
Vg 1 0 Vg
.end";
```

AmpInversor := "Amplificatorul Inversor cu AO\nR1 1 2 R1\nR2 2 3 R2\nE 3 0 0 2 A \
\nVg 1 0 Vg\n.end"

Pentru circuit, calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(AmpInversor,dc,curenti,tensiuni):
Syrup/parsedeck:   Analyzing SPICE deck "Amplificatorul Inversor cu AO"
(ignoring this line)
> tensiuni;
```

$$\left\{ v_2 = \frac{R2\ Vg}{A\,R1 + R1 + R2},\ v_3 = -\frac{A\,R2\ Vg}{A\,R1 + R1 + R2},\ v_1 = Vg \right\}$$

```
> curenti;
```

$$\left\{ i_{R1} = \frac{Vg - \frac{R2\ Vg}{A\,R1 + R1 + R2}}{R1},\ i_{R2} = \frac{\frac{R2\ Vg}{A\,R1 + R1 + R2} + \frac{A\,R2\ Vg}{A\,R1 + R1 + R2}}{R2},\ i_E = \frac{Vg\,(A+1)}{A\,R1 + R1 + R2},\ i_{Vg} = -\frac{Vg\,(A+1)}{A\,R1 + R1 + R2} \right\}$$

Calculul functiei de transfer H(s):

```
> Ha:=eval(v[3]/v[1],tensiuni);
```

$$Ha := -\frac{A\,R2}{R1 + R1\,A + R2}$$

Analiza folosind TTN

Scriem TTN pentru circuitul echivalent al inversorului:

```
> eqTTN:={(Vg-v[2])*1/R1+(v[3]-v[2])*1/R2=0,v[3]=-A*v[2],v[1]=Vg};
```

$$eqTTN := \{\frac{Vg - v_2}{R1} + \frac{v_3 - v_2}{R2} = 0, v_1 = Vg, v_3 = -A\,v_2\}$$

```
> solTTN:=solve(eqTTN,{v[2],v[3],v[1]});
```

$$solTTN := \{v_2 = \frac{R2\,Vg}{A\,R1 + R1 + R2}, v_3 = -\frac{A\,R2\,Vg}{A\,R1 + R1 + R2}, v_1 = Vg\}$$

Functia de transfer:

```
> Ha:=eval(v[3]/v[1],solTTN);
```

$$Ha := -\frac{A\,R2}{R1 + R1\,A + R2}$$

Analiza in cazul ideal

Se considera o comportare in frecventa constanta.
Functia de transfer calculata:

```
> Ha;
```

$$-\frac{A\,R2}{A\,R1 + R1 + R2}$$

Pentru amplificare infinaita relatia se poate aproxima:

```
> H:=limit(Ha,A=infinity);
```

$$H := -\frac{R2}{R1}$$

Evaluare numerica pentru R1=1000, R2=1000 in cele doua cazuri (amplificare infinita si amplificare finita):

```
> Ainfinit:=evalf(eval(H,[R2=10^3,R1=10^3]));
Afinit:=evalf(eval(Ha,[R2=10^3,R1=10^3,A=10^5]));
```

$$Ainfinit := -1.$$

$$Afinit := -.9999800004$$

La intrare aplicam un semnal sinusoidal:

```
> Vg:=sin(w0*t);
```

$$Vg := \sin(w0\,t)$$

La iesire vom avea semnalul de la intrare inversat:

```
> eval(limit(eval(v[3],tensiuni),A=infinity),[R2=10^3,R1=10^3]);
evalf(eval(limit(eval(v[3],tensiuni),A=10^5),[R2=10^3,R1=10^3]))
;
```

$$-\sin(w0\,t)$$

$$-.9999800004\,\sin(w0\,t)$$

Reprezentarea grafica :semnalul de intrare (verde) si semnalul de iesire (rosu):

```
>plot([eval(limit(eval(v[3],tensiuni),A=infinity),[R2=10^3,R1=10^3,w0=2*Pi*10^3]),eval(eval(v[1],tensiuni),[w0=2*Pi*10^3])],t=-0.01..0.01);
```



Analiza in cazul neideal

Se considera o comportare depinzind de frecventa. Pentru amplificatorul operational s-a luat in considerare un singur pol (pol dominant).

```
> A:=A0/(1+s/p1);
```

$$A := \frac{A0}{1 + \frac{s}{p1}}$$

Pentru modelul considerat functia de transfer este:

```
> Ha;
```

$$-\frac{A0\,R2}{\left(1+\frac{s}{p1}\right)\left(\frac{A0\,R1}{1+\frac{s}{p1}} + R1 + R2\right)}$$

Pentru amplificare de cc finita si pentru valorile rezistentelor avem:

```
> Hs:=simplify(eval(Ha, [R2=10^3, R1=10^3, A0=10^5,p1=2*Pi*5*10^3]));
```

$$Hs := -500000000\,\frac{\pi}{500010000\,\pi + s}$$

```
> Bode[castig](evalf(Hs));Bode[faza](evalf(Hs));
```





Amplificarea finita de c.c. a A.O. determina o scadere o amplificarii de c.c. a circ. inveersor.

```
> evalf(eval(Hs,s=I*0));
```

-.9999800004

Analiza SPICE




```
*Amplificator inversor cu AO
.lib "c:\msim62i\lib\jopamp.lib"
R1 in in- 1K
R2 in- out {Rval}
Vcc Vcc 0 10V
Vee Vee 0 -10V
Xopamp 0 in- Vcc Vee out upc741c
Vg in 0 dc 0 ac 100m sin(0 100m 10k)
.param Rval 1k
.step param Rval list 10k 15k 20k
.tran 1u 0.5m
.ac dec 100 0.01 100Meg
.dc Vg -2 2 1m
.probe
.end
```

Functionarea cu limitare

Limitarea in tensiune:



Limitarea in frecventa:



Limitarea unui semnal sinusoidal:



Diagrame Bode de castig si faza:





Topologie neinversoare

Scopul lucrarii

Se doreste analiza circuitului din figura :



Amplificator neinversor cu AO

Componente:

R1=1kO,
R2=12kO;

Calculul functiei de transfer

Metoda I: divizor de tensiune

Calculul functiei de transfer folosind divisor de tensiune:

$$H(s)=\frac{R2+R1}{R1}$$

MetodaII: ecuatii TTN

Pentru circuitul din figura se scrieTTN:

(1) $V_{10}(s)=E(s)$;
(2) $(G_1+G_2)V_{20}(s)-V_{30}(s)=0$;
(3) $V(s)=A(V_{10}(s)-V_{20}(s))$;
ecuatia de iesire este: $V_{30}(s)=Y(s)$;

unde amplificatorul operational s-a modelat ca o sursa de tensiune comandata in tensiune.

In urma rezolvarii acestor ecuatii rezulta functia de transfer:

$$H(s) = \frac{A}{1 + A\dfrac{R1}{R1+R2}}$$

MetodaIII: calcul simbolic

```
> restart:with(Syrup):
> libname:="C:\\maple/SCSlib",libname:
```

Caracterizarea circuitului

Descrierea circuitului folosind un netlist de tip spice

```
> AmpNeinversor:=
"Amplificatorul Neinversor cu AO
R1 0 Inminus R1
R2 Inminus Out R2
E Out 0 In Inminus A
Vg In 0 Vg
.end":
```

Pentru circuit, calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(AmpNeinversor,dc,curenti,tensiuni):
Syrup/parsedeck:   Analyzing SPICE deck "Amplificatorul Neinversor cu AO"
(ignoring this line)
syrup:   There may be an unconnected component.
The following component(s) have zero current: {Vg}.
> tensiuni;
```

$$\{ v_{Out} = \frac{A\,Vg\,(R2 + R1)}{R1 + A\,R1 + R2},\ v_{Inminus} = \frac{A\,Vg\,R1}{R1 + A\,R1 + R2},\ v_{In} = Vg \}$$

```
> curenti;
```

$$\left\{ i_{R1} = -\frac{A\,Vg}{R1 + A\,R1 + R2},\ i_{R2} = \frac{\dfrac{A\,Vg\,R1}{R1 + A\,R1 + R2} - \dfrac{A\,Vg\,(R2 + R1)}{R1 + A\,R1 + R2}}{R2},\ i_E = -\frac{A\,Vg}{R1 + A\,R1 + R2},\ i_{Vg} = 0 \right\}$$

Calculul functiei de transfer H(s):

```
> Ha:=eval(v[Out]/v[In],tensiuni);
```

$$Ha := \frac{A\,(R1 + R2)}{R2 + R1 + A\,R1}$$

Analiza folosind TTN

Scriem TTN pentru circuitul echivalent al inversorului:

```
> restart:with(Syrup):
> libname:="C:\\maple/SCSlib","../DCElib",libname:
> eqTTN:={(v[Out]-v[Inminus])*1/R2+(0-
v[Inminus])*1/R1=0,v[Out]=A*(v[In]-v[Inminus]),v[In]=Vg};
```

$$eqTTN := \{ \frac{v_{Out} - v_{Inminus}}{R2} - \frac{v_{Inminus}}{R1} = 0,\ v_{Out} = A\,(v_{In} - v_{Inminus}),\ v_{In} = Vg \}$$

```
> tensiuni:=solve(eqTTN,{v[Out],v[Inminus],v[In]});
```

$$tensiuni := \{ v_{Out} = \frac{A\, Vg\, (R1 + R2)}{R1\, A + R1 + R2}, v_{In} = Vg, v_{Inminus} = \frac{R1\, A\, Vg}{R1\, A + R1 + R2} \}$$

Functia de transfer:

```
> H:=eval(v[Out]/v[In],tensiuni);
```

$$Ha := \frac{A\,(R1 + R2)}{R2 + R1 + A\, R1}$$

Analiza in cazul ideal

Se considera o comportare in frecventa constanta.
Functia de transfer calculata:

```
> Ha;
```

$$\frac{A\,(R1 + R2)}{R1\, A + R1 + R2}$$

Pentru amplificare infinaita relatia se poate aproxima:

```
> H:=limit(Ha,A=infinity);
```

$$H := \frac{R1 + R2}{R1}$$

Evaluare numerica pentru R1=12000, R2=1000 in cele doua cazuri (amplificare infinita si amplificare finita):

```
> Ainfinit:=evalf(eval(H,[R2=12*10^3,R1=10^3]));
Afinit:=evalf(eval(Ha,[R2=12*10^3,R1=10^3,A=10^5]));
```

$$Ainfinit := 13.$$

$$Afinit := 12.99831022$$

La intrare aplicam un semnal sinusoidal:

```
> eval(v[In],tensiuni);
```

$$\sin(w0\, t)$$

La iesire vom avea semnalul de la intrare amplificat:

```
>
eval(limit(eval(v[Out],tensiuni),A=infinity),[R2=12*10^3,R1=10^3
]);
evalf(eval(limit(eval(v[Out],tensiuni),A=10^5),[R2=12*10^3,R1=10
^3]));
```

$$13\, \sin(w0\, t)$$

$$12.99831022\, \sin(w0\, t)$$

Obs: Diferenta intre a considera o amplificare finita sau una infinita este mica!
Reprezentarea grafica: semnalul de intrarea (verde) si semnalul de iesire (rosu):

```
>
plot([eval(limit(eval(v[Out],tensiuni),A=infinity),[R2=12*10^3,R
1=10^3,w0=2*Pi*10^3]),eval(eval(v[In],tensiuni),[w0=2*Pi*10^3])]
,t=-0.01..0.01);
```



Analiza in cazul neideal

Se considera o comportare depinzind de frecventa. Pentru amplificatorul operational s-a luat in considerare un sinur pol (pol dominant).

```
> A:=A0/(1+s/p1);
```

$$A := \frac{A0}{1+\frac{s}{p1}}$$

Pentru modelul considerat functia de transfer este:

```
> Ha;
```

$$\frac{A0\,(R1+R2)}{\left(1+\frac{s}{p1}\right)\left(\frac{R1\;A0}{1+\frac{s}{p1}}+R1+R2\right)}$$

Pentru amplificare de cc finita si pentru valorile rezestentelor avem:

```
> Hs:=simplify(eval(Ha, [R2=12*10^3, R1=10^3, A0=10^5,
p1=2*Pi*5*10^3]));
```

$$Hs := 13000000000\,\frac{\pi}{1000130000\,\pi+13\,s}$$

```
> Bode[castig](evalf(Hs));Bode[faza](evalf(Hs));
```





Amplificarea finita de c.c. a A.O. determina o scadere o amplificarii de c.c. a circ. neinveersor.

```
> evalf(eval(Hs,s=I*0));
```

12.99831022

Analiza SPICE



Amplificator neinversor cu AO

***Amplificator neinversor cu AO**
.lib "c:\msim62i\lib\jopamp.lib"
R1 0 in- 1K
R2 in- out 10k
Vcc Vcc 0 10V
Vee Vee 0 -10V
Xopamp in+ in- Vcc Vee out upc741c
Vg in+ 0 dc 0 ac 100m sin(0 100m 10k)
.tran 1u 0.5m
.ac dec 100 0.01 100Meg
.dc Vg -2 2 1m
.probe
.end

Functionarea cu limitare

Limitarea in tensiune:



Limitarea in frecventa



Limitarea unui semnal sinusoidal:



Diagrama Bode de cistig si faza:





Modelarea AO

AO in bucla deschisa

```
> restart:with(Syrup):
> circuitAO:=
"amplificator operational in bucla deschisa
Vin In 0
Vcc Vcc 0
Vee Vee 0
Xopamp In 0 Vcc Vee Out uA741
*Modelarea A.O.
.subckt uA741 In_plus In_minus Vcc_plus Vcc_minus Out
V Out 0 f(v[In_plus],v[In_minus],v[Vcc_plus],v[Vcc_minus])
.ends
.end":
> syrup(circuitAO,dc,curenti,tensiuni);
```

Modelarea amplificatorului operational se face la modul general, tensiunea de iesire depinde de tensiunile de intrare (nodul + si nodul -) si de tensiunile de alimentare Vcc si Vee printr-o functie in general neliniara.

```
> v[Out]:=eval(eval(v[Out],tensiuni),f=fsat);
```

$$v_{Out} := \mathrm{fsat}(Vin, 0, Vcc, Vee)$$

Un model simplu este amplificator liniar cu saturatie:

```
> fsat:=(x1,x2,y1,y2)->piecewise(x1-x2<y1/A0 and y2/A0<x1-
x2,A0*(x1-x2),y1/A0<=x1-x2,y1,x1-x2<=y2/A0,y2);
```

$$fsat := (x1, x2, y1, y2) \rightarrow \mathrm{piecewise}\left(x1 - x2 < \frac{y1}{A0} \text{ and } \frac{y2}{A0} < x1 - x2, A0\,(x1 - x2), \frac{y1}{A0} \leq x1 - x2, y1, x1 - x2 \leq \frac{y2}{A0}, y2 \right)$$

```
> #fsat:=(x1,x2,y1,y2)->(y1-y2)/2*tanh(alpha*(x1-x2))+(y1+y2)/2;
```

Pentru conexiunea in bucla deschisa:

```
> Vout:=eval(eval(v[Out],tensiuni),f=fsat);
```

$$Vout := \begin{cases} A0\,Vin & Vin - \frac{Vcc}{A0} < 0 \textbf{ and } \frac{Vee}{A0} - Vin < 0 \\ Vcc & \frac{Vcc}{A0} \leq Vin \\ Vee & Vin \leq \frac{Vee}{A0} \end{cases}$$

Consideram semnalul de intrare sinusoidal:

```
> Vin:=V0*sin(2*Pi*f0*t);
```

$$Vin := V0\,\sin(2\,\pi\,f0\,t)$$

Caracteristica de intrare-iesire cu limitare:

```
>plot([eval(Vin,[V0=0.2,f0=10^3]),subs([V0=0.2,f0=10^3,A0=10^3,Vcc=10,Vee=-10],Vout),t=0..0.001]);
```



Obs: Amplificarea A0 a modelului este constanta. O modelare mai amanuntita poate considera amplificarea depinzind de frecventa.

Limitarea tensiunii de iesire:

```
>plot({eval(10*Vin,[f0=10^3,V0=0.15]),eval(Vout,[f0=10^3,A0=10^2,Vcc=10,Vee=-10,V0=0.15])},t=-0.001..0.001);
```



Topologie inversoare

```
> restart:with(Syrup):
> inversorAO:=
"amplificator operational inversor
Vin In 0
Vcc Vcc 0
Vee Vee 0
R1 In Inm
R2 Inm Out
Xopamp 0 Inm Vcc Vee Out uA741
```

```
*Modelarea A.O.
.subckt uA741 In_plus In_minus Vcc_plus Vcc_minus Out
V Out 0 f(v[In_plus],v[In_minus],v[Vcc_plus],v[Vcc_minus])
.ends
.end":
> sol:=syrup(inversorAO,dc,curenti,tensiuni):
```

Un model simplu este amplificator liniar cu saturatie:

```
> fsat:=(x1,x2,y1,y2)->(y1-y2)/2*tanh(alpha*(x1-x2))+(y1+y2)/2;
```

Tensiunea de iesire:

```
> Vout:=eval(eval(v[Out],tensiuni),f=fsat):
```

Consideram semnalul de intrare sinusoidal:

```
> Vin:=V0*sin(2*Pi*f0*t):
```

Caracteristica de intrare-iesire:

```
>plot([eval(Vin,[V0=10,f0=10^3]),eval(Vout,[V0=10,f0=10^3,Vcc=10,
Vee=-10,alpha=10^3,R1=1,R2=2]),t=0..0.001]);
```



Obs: caracteristica intrare - iesire este cu limitare si corespunde unui amplificator inversor. Pentru functionare liniara, amplificarea este A=-2 si este determinata de rezistentele R1 si R2.

Functionarea liniara (pentru tensiuni de intrare de amplitudine mica V0<Vsat/A):

```
> plot({eval(Vin,[f0=10^3,V0=4]),eval(Vout,[f0=10^3,Vcc=10,Vee=-
10,alpha=10^3,V0=4,R1=1, R2=2])},t=-0.001..0.001);
```

Limitarea tensiunii de iesire (pentru tensiuni de intrare de amplitudine mare V0>Vsat/A):

```
> plot({eval(Vin,[f0=10^3,V0=6]),eval(Vout,[f0=10^3,Vcc=10,Vee=-
10,alpha=10^3,V0=6,R1=1, R2=2])},t=-0.001..0.001);
```





Topologie neinversoare

```
> restart:with(Syrup):
> neinversorAO:=
"amplificator operational neinversor
Vin In 0
```

```
Vcc Vcc 0
Vee Vee 0
R1 0 Inm
R2 Inm Out
Xopamp In Inm Vcc Vee Out uA741
*Modelarea A.O.
.subckt uA741 In_plus In_minus Vcc_plus Vcc_minus Out
V Out 0 f(v[In_plus],v[In_minus],v[Vcc_plus],v[Vcc_minus])
.ends
.end":
> sol:=syrup(neinversorAO,dc,curenti,tensiuni):
```

Un model simplu este amplificator liniar cu saturatie:

```
> fsat:=(x1,x2,y1,y2)->(y1-y2)/2*tanh(alpha*(x1-x2))+(y1+y2)/2;
```

Tensiunea de iesire:

```
> Vout:=eval(eval(v[Out],tensiuni),f=fsat):
```

Consideram semnalul de intrare sinusoidal:

```
> Vin:=V0*sin(2*Pi*f0*t):
```

Caracteristica de intrare-iesire:

```
>plot([eval(Vin,[V0=10,f0=10^3]),eval(Vout,[V0=10,f0=10^3,Vcc=10,
Vee=-10,alpha=10^3,R1=1,R2=2]),t=0..0.001]);
```



Obs: caracteristica intrare - iesire este cu limitare si corespunde unui amplificator neinversor. Pentru functionare liniara, amplificarea este A=3 si este determinata de rezistentele R1 si R2.
Functionarea liniara (pentru tensiuni de intrare de amplitudine mica V0<Vsat/A):

```
> plot({eval(Vin,[f0=10^3,V0=3]),eval(Vout,[f0=10^3,Vcc=10,Vee=-
10,alpha=10^3,V0=3,R1=1, R2=2])},t=-0.001..0.001);
```

Limitarea tensiunii de iesire (pentru tensiuni de intrare de amplitudine mare V0>Vsat/A):

```
> plot({eval(Vin,[f0=10^3,V0=6]),eval(Vout,[f0=10^3,Vcc=10,Vee=-
10,alpha=10^3,V0=6,R1=1, R2=2])},t=-0.001..0.001);
```





Topologie diferentiala

Scopul lucrarii

Se doreste analiza circuitului reprezentind un A.O. conectat in topologie diferentiala ci in figura:



Calculul functiei de transfer

Analiza liniara

Circuitul are doua intrari notate V_{g1} si V_{g2} si o iesire. Fuctionarea schemei este liniara. Pentru a putea calcula semnalul de iesire pastram o singura sursa in intrare si restul le pasivizam:

1) daca V_{g1}=0, atunci vom obtine o structura neinversoare cu un divizor de tensiune R3/R4 rezultand functia de transfer:

$$H_2 = \frac{R_4(R_1 + R_2)}{R_1(R_3 + R_4)}$$

2) daca V_{g2}=0,atunci vom obtine o sructura inversoare rezultand functia de transfer:

$$H_1 = -\frac{R_2}{R_1}$$

Prin suprapunerea efectelor putem calcula semnalul de iesire Y(s):

$$Y(s) = H_1(s)V_{g1}(s) + H_2(s)V_{g2}(s) = -\frac{R_2}{R_1}V_{g1}(s) + \frac{R_4(R_1 + R_2)}{R_1(R_3 + R_4)}V_{g2}(s)$$

sau expresia pentru y(t):

$$y(t) = A_1 \cdot V_{g1}(t) + A_2 \cdot V_{g2}(t) = -\frac{R_2}{R_1} \cdot V_{g1}(t) + \frac{R_4(R_1 + R_2)}{R_1(R_3 + R_4)} \cdot V_{g2}(t)$$

Calcul simbolic

```
> restart:with(Syrup):
> libname:="C:\\maple/SCSlib",libname:
```

Caracterizarea circuitului

Descrierea circuitului folosind un netlist de tip spice

```
> Amp:=
"Amplificatorul cu AO
Vg1 1 0
Vg2 3 0
R1 1 2
R2 2 5
R3 3 4
R4 4 0
E 5 0 4 2 A
.end":
```

Pentru circuit, calculul tensiunilor nodale si a curentilor prin laturi

```
> syrup(Amp,ac,curenti,tensiuni);
```

$$\{v_5 = -\frac{A\,(-R1\;R4\;Vg2 + Vg1\;R4\;R2 - R2\;R4\;Vg2 + Vg1\;R3\;R2)}{R4\;R2 + R4\;R1 + R4\;R1\;A + R3\;R2 + R3\;R1 + R3\;R1\;A},$$
$$v_2 = \frac{Vg1\;R4\;R2 + Vg1\;R3\;R2 + R1\;R4\;A\;Vg2}{R4\;R2 + R4\;R1 + R4\;R1\;A + R3\;R2 + R3\;R1 + R3\;R1\;A},\; v_4 = \frac{R4\;Vg2}{R3 + R4},\; v_3 = Vg2,$$
$$v_1 = Vg1\}$$

Tensiunea de iesire este:

```
> Y:=collect(factor(eval(v[5],tensiuni)),{Vg1,Vg2});
```

$$Y := -\frac{A\,(R4\;R2 + R3\;R2)\,Vg1}{(R3 + R4)\,(R2 + R1 + R1\;A)} - \frac{A\,(-R4\;R1 - R4\;R2)\,Vg2}{(R3 + R4)\,(R2 + R1 + R1\;A)}$$

Pentru amplificare infinita:

```
> Y1:=collect(factor(limit(Y,A=infinity)),{Vg1,Vg2});
```

$$Y1 := -\frac{(R4\;R2 + R3\;R2)\,Vg1}{(R3 + R4)\,R1} - \frac{(-R4\;R1 - R4\;R2)\,Vg2}{(R3 + R4)\,R1}$$

Functii de transfer

Formula pentru calculul iesirii: Y(s)=H1(s)Vg1(s)+H2(s)Vg2(s) unde

```
> H1:=limit(eval(v[5]/v[1],tensiuni), Vg2=0);
```

$$H1 := -\frac{R2\;A}{R2 + R1 + R1\;A}$$

```
> H2:=factor(limit(eval(v[5]/v[3],tensiuni), Vg1=0));
```

$$H2 := \frac{A\;R4\;(R1 + R2)}{(R3 + R4)\,(R2 + R1 + R1\;A)}$$

Particularizari

Amplificator inversor

Schema inversorului:



Functia de transfer calculata:

> **H1;**

$$-\frac{R2\ A}{R2 + R1 + R1\ A}$$

Pentru amplificare infinaita relatia se poate aproxima:

> **limit(H1,A=infinity);**

$$-\frac{R2}{R1}$$

Amplificator neinversor

Schema neinversorului:



Functia de transfer calculata:

> **H2;**

$$\frac{A\ R4\ (R1 + R2)}{(R3 + R4)\ (R2 + R1 + R1\ A)}$$

Pentru amplificare infinaita relatia se poate aproxima:

> **limit(H2,A=infinity);**

$$\frac{R4\ (R1 + R2)}{(R3 + R4)\ R1}$$

Amplificator diferential

Shema montajului diferential:



Functia de transfer calculata:

```
> H:=simplify(subs({Vg1=Vg/2, Vg2=-Vg/2},Y)/Vg);
```

$$H := -\frac{1}{2}\frac{A(2\,R4\,R2 + R3\,R2 + R4\,R1)}{(R3 + R4)(R2 + R1 + R1\,A)}$$

Pentru amplificare infinaita relatia se poate aproxima:

```
> limit(H,A=infinity);
```

$$-\frac{1}{2}\frac{2\,R4\,R2 + R3\,R2 + R4\,R1}{(R3 + R4)\,R1}$$

O schema simplificata pentru amplificatorul diferential are rezistentele egale R3=R1, R4=R2.
In acest caz amplificarea este:

```
> H:=simplify(subs({Vg1=Vg/2, Vg2=-Vg/2, R3=R1, R4=R2},Y)/Vg);
```

$$H := -\frac{R2\,A}{R2 + R1 + R1\,A}$$

Pentru amplificare infinaita relatia se poate aproxima:

```
> limit(H,A=infinity);
```

$$-\frac{R2}{R1}$$